

# ***Lecture 8***

## ***The B Method***

***Refining B Specifications - Data Refinement***

# Lecture Outline

---

- Refinement in B
- Arrays
- Sequential Composition
- Local Variables
- Data Refinement
- Inheriting Static Information
- Including and Seeing Machines in Refinements

## References

---

- [1] Abrial, J.-R., *The B Book - Assigning Programs to Meanings*, Cambridge University Press, 1996. (chapter 11)
- [2] Schneider, S., *The B-Method - An Introduction*, Palgrave Macmillan, Cornerstones of Computing series, 2001. (chapter 12)
- [3] Clearsy System Engineering, *AtelierB home page*  
<http://www.atelierb.eu/en/>
- [4] Clearsy System Engineering, *B Method home page*  
<http://www.methode-b.com/en/>

# Refinement in B

---

- *Refinement* enables evolving an abstract system specification into an executable implementation
- B supports the concept of *stepwise refinement*
  - The implementation is developed gradually, through a series of *refinement steps*, each introducing a small number of design decisions
  - Intermediate steps between specification and implementation contain both abstract constructs and implementation details and are described by means of *refinement machines*
- Refinement topics
  - **Refinement of data**
  - Refinement of nondeterminism
  - Refinement proof obligations

# Arrays

- An *array* is a named, indexed collection of values of a given type
- An array  $a$  of size  $n$  is a mapping (function) from  $1 \dots n$  to the type of values contained by the array
  - E.g. the array  $a : 3 \ 15 \ 7 \ 19$  can be thought of as the function  $a : 1 \dots 4 \rightarrow \mathbb{N}$ ,  $a = \{1 \mapsto 3, 2 \mapsto 15, 3 \mapsto 7, 4 \mapsto 19\}$
- Round brackets are used to provide access to elements of an array (as opposed to square brackets used in programming languages)
  - $a(i)$  = the  $i$ -th element of  $a$ , or the function  $a$  applied to  $i$
- AMN allows assigning values to indexed elements of an array
  - $a(i) := e$ , shorthand for  $a := a \leftarrow \{i \mapsto e\}$

# Sequential Composition

---

- *Sequential composition* allows one substitution to be executed after another
- Construct:  $S; T$ , where  $S$  and  $T$  are substitutions
- Definition (by post-condition establishment):  $[S; T]P \Leftrightarrow [S]([T]P)$
- Any number of substitutions can be sequentially composed
- E.g.

$$\begin{aligned}[x := x + 2; x := x + 4](x > 9) &= [x := x + 2](x + 4 > 9) \\ &= x + 2 + 4 > 9 \\ &= x > 3\end{aligned}$$

$$\begin{aligned}[t := x; x := y; y := t](x > 6 \wedge y < 4) &= [t := x; x := y](x > 6 \wedge t < 4) \\ &= [t := x](y > 6 \wedge t < 4) \\ &= y > 6 \wedge x < 4\end{aligned}$$

## Local Variables

- In the context of sequential composition, *local variables* are useful for accomplishing particular computations without imposing on the overall state space
- Construct:  $\text{VAR } t \text{ IN } S \text{ END}$ , where  $t$  is a variable and  $S$  is a substitution
- $S$  should initialize  $t$  before using it
- E.g.  $\text{VAR } t \text{ IN } t := x; x := y; y := t \text{ END}$
- Lists of several local variables can be described this way, if needed
- Definition (by post-condition establishment):  
$$[ \text{VAR } t \text{ IN } S \text{ END} ] P \Leftrightarrow \forall t \cdot [S]P$$
- E.g. 
$$\begin{aligned} [ \text{VAR } t \text{ IN } t := x; x := y; y := t \text{ END} ] (x = A \wedge y = B) \\ &= \forall t \cdot [t := x; x := y; y := t](x = A \wedge y = B) \\ &= \forall t \cdot (y = A \wedge x = B) \\ &= (y = A \wedge x = B) \end{aligned}$$

# Data Refinement

---

- Abstract machines describe the state of the system in terms of abstract math structures (sets, relations, functions, sequences)
  - *Users* should only perceive the behavior of the system in terms of this specification
- System implementors though are concerned with data representation, since the abstract specification constructs are not directly supported by conventional programming languages
- Data representation is provided by means of a *refinement machine*, that captures:
  - the refined state
  - the connection between abstract and refined state by means of a linking invariant
  - the way that the initialisation and operations work with the new data representation
- A refinement machine has the same interface as the machine it refines (same operations with the same signatures)



## Data Refinement (cont.)

- E.g.: specification of a football team
  - Keeps track of players in the field within a game
  - The team starts with the first 11 players
  - Operations are provided to replace a player on the field and query whether a particular team member is currently playing or not

```
MACHINE Team
SETS ANSWER = {in, out}
VARIABLES team
INVARIANT  $team \subseteq 1 \dots 22 \wedge \text{card}(team) = 11$ 
INITIALISATION team := 1 .. 11
OPERATIONS
  substitute(pp, rr)  $\hat{=}$ 
    PRE  $pp \in team \wedge rr \in 1 \dots 22 \wedge rr \notin team$ 
    THEN  $team := (team - \{pp\}) \cup \{rr\}$ 
    END;

  aa  $\leftarrow$  query(pp)  $\hat{=}$ 
    PRE  $pp \in 1 \dots 22$ 
    THEN
      IF  $pp \in team$ 
      THEN  $aa := in$ 
      ELSE  $aa := out$ 
      END
    END
END
```

## Data Refinement (cont.)

- Refinement for *Team*
  - Uses an array of size 11 to represent the state information
  - An abstract state may have several corresponding concrete states
  - Refined operations are assumed to work within their preconditions, as given in the abstract machine

```
REFINEMENT TeamR1
REFINES Team
VARIABLES teamr
INVARIANT  $teamr \in 1 \dots 11 \mapsto 1 \dots 22 \wedge$ 
            $\text{ran}(teamr) = team$ 
INITIALISATION  $teamr := \text{id}(1 \dots 11)$ 
OPERATIONS
  substitute ( pp , rr )  $\hat{=}$ 
  BEGIN
     $teamr(teamr^{-1}(pp)) := rr$ 
  END;

  aa  $\leftarrow$  query ( pp )  $\hat{=}$ 
  IF  $pp \in \text{ran}(teamr)$ 
  THEN aa := in
  ELSE aa := out
  END
END
```

## Data Refinement (cont.)

- Alternative refinement for Team
  - Uses an array indexed by team members, rather than places in the team
  - Uses sequential composition to define one the operations

```
REFINEMENT TeamR2
REFINES Team
VARIABLES teama
INVARIANT  $teama \in 1 \dots 22 \rightarrow ANSWER \wedge$ 
            $team = (teama^{-1})[\{in\}]$ 
INITIALISATION
   $teama := (1 \dots 11) \times \{in\} \cup (12 \dots 22) \times \{out\}$ 
OPERATIONS
  substitute( $pp, rr$ )  $\hat{=}$ 
    BEGIN
       $teama(pp) := out;$ 
       $teama(rr) := in$ 
    END;

   $aa \leftarrow query(pp)$   $\hat{=}$ 
    BEGIN
       $aa := teama(pp)$ 
    END

END
```

# Inheriting Static Information

---

- A refinement has access to all the static information of the abstract machine it refines (parameters, sets, constants)
- A refinement may also access the sets and constants of any machine included in the abstract machine it refines
- A refinement has no default access to the information contained in machines seen/used by the abstract machine it refines; access to such information may be provided if a corresponding SEES clause is explicitly inserted in the refinement machine

## Inheriting Static Information (cont.)

- E.g.: specification of a stack pile, used to track items of clothing to be ironed
  - Has a parameter, *limit*, which is the maximum number of items that can be piled
  - Has a given set, *ITEM*, defining the items that can be ironed

```
MACHINE Ironing(limit)
CONSTRAINTS limit ∈ ℕ1
SETS ITEM
VARIABLES pile
INVARIANT pile ∈ seq(ITEM) ∧ size(pile) ≤ limit
INITIALISATION pile := []
OPERATIONS

  put(ii) ≡
    PRE ii ∈ ITEM ∧ size(pile) < limit
    THEN pile := pile ← ii
    END;

  ii ← take ≡
    PRE pile ≠ []
    THEN pile := front(pile) || ii := last(pile)
    END;

  bb ← query(ii) ≡
    PRE ii ∈ ITEM
    THEN bb := bool(ii ∈ ran(pile))
    END

END
```

## Inheriting Static Information (cont.)

- Refinement machine for `Ironing`
  - Uses an array to store the items to be ironed
  - Uses a counter variable to indicate how much of the array corresponds to the abstract state

```
REFINEMENT IroningR (limit)  
REFINES Ironing  
VARIABLES pilearr, counter  
INVARIANT  $pilearr \in 1 \dots limit \rightarrow ITEM \wedge counter \in 0 \dots limit \wedge$   
            $1 \dots counter \triangleleft pilearr = pile$   
INITIALISATION pilearr :=  $\emptyset$  || counter := 0  
OPERATIONS  
  put(ii)  $\hat{=}$   
  BEGIN  
    counter := counter + 1;  
    pilearr(counter) := ii  
  END;  
  
  ii  $\leftarrow$  take  $\hat{=}$   
  BEGIN  
    ii := pilearr(counter);  
    counter := counter - 1  
  END;  
  
  bb  $\leftarrow$  query(ii)  $\hat{=}$   
  IF ii  $\in$  pilearr[1 .. counter]  
  THEN bb := TRUE  
  ELSE bb := FALSE  
  END  
  
END
```

# Including and Seeing Machines in Refinements

---

- Refinement machines may be incrementally developed using the structuring mechanisms provided by B
  - Abstract machines may be included/extended and operations promoted in refinements (similar to inclusion in other abstract machines)
  - Abstract machines can be seen by refinements
  - The `USES` clause cannot appear in a refinement (therefore just `INCLUDES`, `PROMOTES`, `EXTENDS` and `SEES`)
- The linking invariant of a refinement links all its state (native and included) with the corresponding abstract state
- The included state can be used in read mode in the description of operations of the refinement machine, but can only be modified through calls of operations of the included/extended machines (not by direct assignment)



## Including and Seeing Machines in Refinements (cont.)

- E.g.: Specification of a ship control system in a port

```
MACHINE Port
SETS SHIP, QUAJ
VARIABLES waiting, docked
INVARIANT waiting ∈ iseq(SHIP) ∧ docked ∈ QUAJ ↔ SHIP ∧
    ran(waiting) ∩ ran(docked) = ∅
INITIALISATION waiting := [] || docked := ∅
OPERATIONS
  arrive(ss) ≡
    PRE ss ∈ SHIP ∧ ss ∉ ran(waiting) ∧ ss ∉ ran(docked)
    THEN waiting := waiting ← ss
    END;

  dock(qq) ≡
    PRE waiting ≠ [] ∧ qq ∈ QUAJ ∧ qq ∉ dom(docked)
    THEN waiting := tail(waiting) || docked(qq) := first(waiting)
    END;

  qq ← leave(ss) ≡
    PRE ss ∈ SHIP ∧ ss ∈ ran(docked)
    THEN docked := docked ⊖ {ss} || qq := docked-1(ss)
    END;

  nn ← numberwaiting ≡
    BEGIN
      nn := size(waiting)
    END

END
```



## Including and Seeing Machines in Refinements (cont.)

- Machine `Port` will be refined by including two simpler machines
  - `List` - for handling the queue of waiting ships
  - `Map` - for handling the docking of ships to quays

```
MACHINE List(ELEMENT)
VARIABLES list
INVARIANT list ∈ seq(ELEMENT)
INITIALISATION list := []
OPERATIONS
  add(ee) ≡
    PRE ee ∈ ELEMENT
    THEN list := list ← ee
    END;

  ee ← take ≡
    PRE list ≠ []
    THEN list := tail(list) || ee := front(list)
    END

END
```

```
MACHINE Map(INDEX, ITEM)
VARIABLES fun
INVARIANT fun ∈ INDEX → ITEM
INITIALISATION fun := ∅
OPERATIONS
  insert(ss1, ss2) ≡
    PRE ss1 ∈ INDEX ∧ ss2 ∈ ITEM
    THEN fun(ss1) := ss2
    END;

  remove(ss1) ≡
    PRE ss1 ∈ INDEX
    THEN fun := {ss1} ↖ fun
    END;

  ss2 ← query(ss1) ≡
    PRE ss1 ∈ dom(fun)
    THEN ss2 := fun(ss1)
    END

END
```

# Including and Seeing Machines in Refinements (cont.)

- Refinement of Port with inclusion

```
REFINEMENT PortR
REFINES Port
INCLUDES List(SHIP), Map(SHIP, QUAI)
VARIABLES num
INVARIANT waiting = list  $\wedge$  docked-1 = fun  $\wedge$  num = size(waiting)
INITIALISATION num := 0
OPERATIONS
  arrive(ss)  $\hat{=}$ 
  BEGIN
    add(ss);
    num := num + 1
  END;

  dock(qq)  $\hat{=}$ 
  BEGIN
    VAR sh IN
      sh  $\leftarrow$  take;
      insert(sh, qq)
    END;
    num := num - 1
  END;

  qq  $\leftarrow$  leave(ss)  $\hat{=}$ 
  BEGIN
    qq  $\leftarrow$  query(ss);
    remove(ss)
  END;

  nn  $\leftarrow$  numberwaiting  $\hat{=}$ 
  num := num

END
```